Letter to the Editor

Sir:

The determination of the chemical selectivity (Sl and Sln) and activity of a catalyst used to hydrogenate a vegetable oil is a recommended practice method in the AOCS Standard Methods (1).

To determine the Sl and Sln, a sample of soybean oil is hydrogenated and from the fatty acid composition of the starting and ending samples the Sl and Sln may be calculated. The calculation may be made by a graphical method, a computer programmed with a program in Fortran IV or a good calculator (2).

With the advent and wide distribution of the low cost "home computer," it is possible to have available a computer to do the calculations necessary for the selectivity determinations. Since most low cost computers use BASIC programming, and only a Fortran IV program is available for the selectivity calculation, this letter describes a BASIC program for the calculations.

The program was written for a TRS80-Color Computer, 16K with extended BASIC, so some statements (e.g., the print statements, lines 630-710) would need to be changed for your particular computer.

In the program, lines 20-30 are REM statements for description only. Line 40 reads the data in the order: I = sample designation, SO = start oleic, SL = start linoleic and S3 = start linolenic, FO = final oleic, FL = final linoleic, F3 = final linolenic and AT is reaction time.

Lines 50-80 direct the calculation if no linolenic is found in the starting or ending composition.

Line 90 calculates K1 from the data. The equations used were described by Albright (3). The calculation of K1 has a closed solution.

Lines 120-260 are the calculation of K2. Since there is not a closed solution for K2, an iteration is used to find the K2 that will give the final linoleic. A counter (lines 130 and 205) is used to stop the iteration after 50 times if the solution is not found.

Lines 160 and 170 are used if there is linolenic in the starting sample but not the final sample. The K1 is then assumed to be 2.3 X K2. This is valid only if a nickel catalyst is used since we found the Sln for nickel catalysts is usually ca. 2.3 (4).

Lines 280-420 are the calculation of K3. Again an iterative procedure is used with a counter to stop the iteration after 50 trials.

Lines 430-580 are the calculation of K2 and K3 if no linolenic was in the starting sample. Thus the linoleic selectivity may be calculated using oils such as cottonseed, peanut, corn, etc.

Lines 600-620 are the calculation of the decrease in iodine value during the hydrogenation and the IV drop per minute is calculated and printed.

Lines 630-700 are statements to print on a printer. If a printer is not available, these would need to be printed on the screen and copied.

Line 1000 starts the Data section. The data is listed in the same order as the Read statement (line 40). As many as 999 (lines 1000-1998) problems or to the capacity of the memory available in the particular computer may be entered at once.

The use of this BASIC program in a small computer produces the actual values of the three reaction rate constants, as well as the Sl and Sln and the IV drop/min.

The K values may be used to regenerate the complete hydrogenation curves by substituting them back into the equations and calculating the composition at anytime.

ROBERT R. ALLEN
Rt. 9, Box 137
McKinney, TX 75069

REFERENCES

1. Official and Tentative Methods of the American Oil Chemists' Society, AOCS, Champaign, IL, 1973, Method TZ 16-79.
2. Allen, R.R., JAOCS 59:204 (1982).
3. Albright, L.F., JAOCS 42:250 (1965).
4. Allen, R.R., JAOCS 58:166-9 (1981).

BASIC Program for the Determination of Sl and Sln

```
10 CLS
20 'CALCULATION OF K1,K2,K3,Sl,Sln,IV/MIN
30 'PROGRAM"SELECTIVITY",R.R. ALLEN,7/5/83
40 READ I,SO,SL,S3,FO,FL,F3,AT
50 IF I=0 THEN 2000
60 PRINT"SAMPLE #";I
70 IF S3=0 THEN 430
80 IF F3=0 THEN 120
90 'CALCULATION OF K1
100 K1=(-LOG(F3/S3))/AT
110 PRINT"K1=";K1:PRINT
120 'CALCULATION OF K2
130 C1=0
140 A2=0:A1=2.0
150 K2=A1+(A2-A1)/2
160 IF F3=0 THEN 170 ELSE 180
170 K1=2.3*K2
180 AA=K1/(K2-K1):FK=EXP(-K2*AT):EK=EXP(-K1*AT)
190 F2=S3*AA*(EK-FK)+SL*FK
200 DL=F2-FL
205 C1=C1+1:IF C1-50=0 THEN 206 ELSE 210
206 PRINT"K2 WILL NOT CONVERGE":GOTO 40
210 IF ABS(DL)-.0001<=0 THEN 270 ELSE 220
220 IF DL<0 THEN 250 ELSE 260
250 A1=K2: GOTO 150
260 A2=K2: GOTO 150
270 PRINT"K2=";K2: PRINT
280 'CALCULATION OF K3
290 C3=0
```

```
300 A2=0: A1=1
310 K3=A1+(A2-A1)/2
320 AB=K2/(K3-K1):GK=EXP(-K3*AT):AC=K2/(K3-K2)
330 F1=S3*AA*AB*(EK-GK)-83*AA*AC*(FK-GK)+SL*AC*(FK-GK)+SO*GK
340 C3=C3+1:IF C3-50=0 THEN 350 ELSE 360
350 PRINT"K3 WILL NOT CONVERGE":GOTO 40
360 D1=F1-FO
370 IF ABS(D1)-.0001 <=0 THEN 410 ELSE 380
380 IF D1<0 THEN 390 ELSE 400
390 A1=K3:GOTO 310
400 A2=K3:GOTO 310
410 PRINT"K3=";K3:PRINT
420 GOTO 600
430 'CALCULATE K2 IF NO LINOLENIC PRESENT
435 K1=0
440 K2=-LOG(FL/SL)/AT
450 PRINT "K2=";K2:PRINT
460 'CALCULATE K3 IF NO LINOLENIC
470 C2=0
480 A2=0 :A1=2.0
490 K3=A1+(A2-A1)/2
500 AA=K2/(K3-K2): FK=EXP(-K3*AT):EK=EXP(-K2*AT)
510 AO=SL*AA*(EK-FK)+SO*FK
520 OD=AO-FO
530 IF ABS(OD)-.0001<=0 THEN 590
540 C2=C2+1:IF C2-50=0 THEN 550 ELSE 560
550 PRINT"K3 WILL NOT CONVERGE":GOTO 40
560 IF OD<0 THEN 570 ELSE 580
570 A1=K3:GOTO 490
580 A2=K3:GOTO 490
590 PRINT "K3=";K3:PRINT
600 SV=2.616*S3+1.732*SL+.8601*SO
610 FV=2.616*F3+1.732*FL+.8601*FO
620 IV=SV-FV
630 'PRINT STATEMENTS
640 PRINT#-2,"SAMPLE #"; I
650 PRINT#-2,"K1="K1;"     K2=";K2;"     K3=";K3
660 PRINT#-2,""
670 PRINT#-2,"     Sl=";K2/K3;"     Sln=";K1/K2
680 PRINT#-2
690 PRINT#-2,"START IV=";SV;"     FINAL IV =";FV
700 PRINT#-2,"          IV DROP PER MIN. IS";IV/AT
710 PRINT#-2:PRINT#-2:PRINT#-2
1000 DATA 1,44.74,33.45,0,57.5,17.95,0,16.5
1001 DATA 2,24.99,52.01,6.8,68.42,14.29,.53,43
1999 GOTO 40
2000 END
```

```
SAMPLE # 1
K1= 0        K2= .0377249498        K3= 3.19504738E-03

      Sl= 11.8073209        Sln= 0

START IV= 96.416274        FINAL IV = 80.54515
          IV DROP PER MIN. IS .961886301


SAMPLE # 2
K1= .0593442066        K2= .034371376        K3= 2.53677368E-04

      Sl= 135.492481        Sln= 1.72655894

START IV= 129.364019        FINAL IV = 84.984802
          IV DROP PER MIN. IS 1.03207481
```

ERRATUM

In the article "The Freeze Fracture Ultrastructure of Peanut Oil and Other Natural and Synthetic Triacylglycerol Droplets" appearing in the July issue of *JAOCS* (Rigler, Roth, Kritchevsky and Patton 60:1291 [1983]), a number of freeze fracture electron micrographs were included among the figures. To show the micrographs in greater detail, they are reproduced, larger, on the following pages.